// Read in scraped bill categories etc.
// Scraped by Valeron from here:
// https://www.althingi.is/thingstorf/listar-yfir-mal-a-yfirstandandi-thingi/thingmal-eftir-efnisflokkum/
cd "$path_pch"
pause on

import excel using Data\raw\Voting\bill_type/bill_categories, clear first

ren SessionID session_id
assert inrange(session_id,127,149)

ren Category billcat_ice
ren Category_1 billcat_eng
ren Málsnúmer billnum
ren Dagsetning billdate
ren Heitimáls billtitle_ice
ren NameofCase billtitle_eng
drop Flutningsmaður
ren Mover mover

g billgencat_ice = ""

replace billgencat_ice="Atvinnuvegir" if inlist(billcat_ice,"Ferðaþjónusta","Iðnaður","Landbúnaður","Sjávarútvegur","Tölvu- og upplýsingamál","Viðskipti")
replace billgencat_ice="Erlend samskipti" if inlist(billcat_ice,"Alþjóðasamningar og utanríkismál","Norræn málefni")
replace billgencat_ice="Hagstjórn" if inlist(billcat_ice,"Efnahagsmál","Fjárreiður ríkisins","Skattar og tollar")
replace billgencat_ice="Heilsa og heilbrigði" if inlist(billcat_ice,"Heilbrigðiseftirlit","Heilbrigðismál")
replace billgencat_ice="Lög og réttur" if inlist(billcat_ice,"Dómstólar og réttarfar","Löggæsla og eftirlit","Persónuleg réttindi")
replace billgencat_ice="Mennta- og menningarmál" if inlist(billcat_ice,"Íþróttir og æskulýðsmál","Menningarmál","Menntamál")
replace billgencat_ice="Samfélagsmál" if inlist(billcat_ice,"Almannatryggingar","Atvinnumál","Byggðamál","Félagsmál")
replace billgencat_ice="Samgöngumál" if inlist(billcat_ice,"Fjarskipti og póstmál","Samgöngur")
replace billgencat_ice="Stjórnarskipan og stjórnsýsla" if inlist(billcat_ice,"Alþingi","Stjórnkerfi og stjórnarskipunarmál","Sveitarstjórnarmál")
replace billgencat_ice="Trúmál og kirkja" if inlist(billcat_ice,"Trúfélög og trúarbrögð","Þjóðkirkjan")
replace billgencat_ice="Umhverfismál" if inlist(billcat_ice,"Mengun","Orkumál og auðlindir","Umhverfisstjórn og náttúruvernd")

g billgencat_eng = ""

replace billgencat_eng="Industry" if inlist(billcat_ice,"Ferðaþjónusta","Iðnaður","Landbúnaður","Sjávarútvegur","Tölvu- og upplýsingamál","Viðskipti")
replace billgencat_eng="Foreign Relations" if inlist(billcat_ice,"Alþjóðasamningar og utanríkismál","Norræn málefni")
replace billgencat_eng="Economic Management" if inlist(billcat_ice,"Efnahagsmál","Fjárreiður ríkisins","Skattar og tollar")
replace billgencat_eng="Health" if inlist(billcat_ice,"Heilbrigðiseftirlit","Heilbrigðismál")
replace billgencat_eng="Law and Justice" if inlist(billcat_ice,"Dómstólar og réttarfar","Löggæsla og eftirlit","Persónuleg réttindi")
replace billgencat_eng="Education and Culture" if inlist(billcat_ice,"Íþróttir og æskulýðsmál","Menningarmál","Menntamál")
replace billgencat_eng="Community Issues" if inlist(billcat_ice,"Almannatryggingar","Atvinnumál","Byggðamál","Félagsmál")
replace billgencat_eng="Transport" if inlist(billcat_ice,"Fjarskipti og póstmál","Samgöngur")
replace billgencat_eng="Administration and Local Governance" if inlist(billcat_ice,"Alþingi","Stjórnkerfi og stjórnarskipunarmál","Sveitarstjórnarmál")
replace billgencat_eng="Religion" if inlist(billcat_ice,"Trúfélög og trúarbrögð","Þjóðkirkjan")
replace billgencat_eng="Environment" if inlist(billcat_ice,"Mengun","Orkumál og auðlindir","Umhverfisstjórn og náttúruvernd")

assert !mi(billcat_eng) & !mi(billcat_ice) & !mi(billgencat_eng) & !mi(billgencat_ice)

g billgencat_engshort = ""

replace billgencat_engshort="industry" if billgencat_eng=="Industry"
replace billgencat_engshort="foreign" if billgencat_eng=="Foreign Relations"
replace billgencat_engshort="econ" if billgencat_eng=="Economic Management"
replace billgencat_engshort="health" if billgencat_eng=="Health"
replace billgencat_engshort="law" if billgencat_eng=="Law and Justice"
replace billgencat_engshort="educ" if billgencat_eng=="Education and Culture"
replace billgencat_engshort="community" if billgencat_eng=="Community Issues"
replace billgencat_engshort="transport" if billgencat_eng=="Transport"
replace billgencat_engshort="admin" if billgencat_eng=="Administration and Local Governance"
replace billgencat_engshort="relig" if billgencat_eng=="Religion"
replace billgencat_engshort="environ" if billgencat_eng=="Environment"

assert !mi(billgencat_engshort)

order session_id

// how many categories per bill?
bys session_id billnum: g numCat = _N

preserve
	keep session_id billnum numCat
	duplicates drop
	tab numCat
restore

// convert to one observation per bill, with dummies for being in different categories
foreach x in industry foreign econ health law educ community transport admin relig environ {
	g `x'_cattmp = billgencat_engshort=="`x'"
	bys session_id billnum: egen `x'_cat = max(`x'_cattmp)
	drop `x'_cattmp
}

keep session_id billnum *_cat
duplicates drop
isid session_id billnum
ren billnum bill_id

save Data/intermediate/Voting/bill_type/bill_cat_clean, replace

